> Soluzioni del compito di Calcolo Numerico del 25 ottobre 2004

> **Enrico Bertolazzi**

Esercizio 1

```
> restart:
  with(LinearAlgebra) :
  with(plots):
  with(Student[Calculus1]):
```

```
Warning, the name changecoords has been redefined
```

Primitiva della funzione da integrare

```
> F := cos(x^2) ;
```

$$F := \cos(x^2)$$

Funzione da integrare

```
> f := expand(diff(F,x)) ;
```

$$f := -2 \sin(x^2)\, x$$

Derivata quarta

```
> ddddf := expand(diff(f,x,x,x,x)) ;
```

$$ddddf := -32 \sin(x^2)\, x^5 + 160 \cos(x^2)\, x^3 + 120 \sin(x^2)\, x$$

Studio della derivata quarta

```
> plot( abs(ddddf), x=-1..2 ) ;
```



stima del modulo della derivata quarta

```
> M := 32*2^5+160*2^3+120*2;
```

$$M := 2544$$

stima degli intervalli necessari col metodo di Simpson per avere E < 0.001

```
> a,b,E := -1,2,0.001 ;
  expr  := E - (b-a)^5/(180*n^4)*M ;
```

$$a, b, E := -1, 2, 0.001$$

$$expr := 0.001 - \frac{17172}{5\, n^4}$$

```
> expr1 := isolate( expr, n^4 ) ;
```

$$expr1 := n^4 = 3.434400000\ 10^6$$

```
> nint := (op(2,expr1))^(1/4) ;
```

$$nint := 43.04896544$$

```
> ApproximateInt(f, x=-1..2, method = simpson, partition = 16, output =
  plot);
```




```
> evalf(ApproximateInt(f, x=-1..2, method = simpson, partition = 6));
```

-1.202497505

```
> esatto := evalf(subs(x=2,F)-subs(x=-1,F)) ;
```

$esatto := -1.193945927$

Esercizio 2

```
> restart:
  with(LinearAlgebra) :
```

Soluzione del problema di interpolazione

```
> psol := x -> 3*x-x^3;
```

$$psol := x \to 3\,x - x^3$$

Punti di interpolazione:

```
> X := [-2,0,1,2,5];
  Y := [seq(psol(X[i]),i=1..5)];
```

$$X := [-2, 0, 1, 2, 5]$$

$$Y := [2, 0, 2, -2, -110]$$

Polinomio interpolante:

```
> interp( X, Y, 'z');
```

$$-z^3 + 3z$$

Costruzione delle differenze divise di ordine 0

```
> f1 := Y[1];
  f2 := Y[2];
  f3 := Y[3];
  f4 := Y[4];
  f5 := Y[5];
```

$$f1 := 2$$
$$f2 := 0$$
$$f3 := 2$$
$$f4 := -2$$
$$f5 := -110$$

Differenze divise

```
> f12 := (f2-f1)/(X[2]-X[1]);
  f23 := (f3-f2)/(X[3]-X[2]);
  f34 := (f4-f3)/(X[4]-X[3]);
  f45 := (f5-f4)/(X[5]-X[4]);
```

$$f12 := -1$$
$$f23 := 2$$
$$f34 := -4$$
$$f45 := -36$$

Differenze divise seconde

```
> f123 := (f23-f12)/(X[3]-X[1]);
  f234 := (f34-f23)/(X[4]-X[2]);
  f345 := (f45-f34)/(X[5]-X[3]);
```

$$f123 := 1$$
$$f234 := -3$$
$$f345 := -8$$

Differenze divise terze

```
> f1234 := (f234-f123)/(X[4]-X[1]);
  f2345 := (f345-f234)/(X[5]-X[2]);
```

$$f1234 := -1$$
$$f2345 := -1$$

Differenze divise quarte

```
> f12345 := (f2345-f1234)/(X[5]-X[1]);
```

$$f12345 := 0$$

Polinomi della base

```
> w0 := 1 ;
```

```
w1 := x-X[1] ;
w2 := expand(w1 * ( x - X[2])) ;
w3 := expand(w2 * ( x - X[3])) ;
w4 := expand(w3 * ( x - X[4])) ;
```

$$w0 := 1$$

$$w1 := x + 2$$

$$w2 := x^2 + 2\,x$$

$$w3 := x^3 + x^2 - 2\,x$$

$$w4 := x^4 - x^3 - 4\,x^2 + 4\,x$$

Polinomio interpolante

```
> p := f1*w0 + f12 * w1 + f123 * w2 + f1234 * w3 + f12345 * w4 ;
```

$$p := 3\,x - x^3$$

Esercizio 3

```
> restart:
  with(LinearAlgebra) :
```

Definizione del problema

```
> p := x -> 1-x^2 ;
  q := x -> -1 ;
  r := x -> x - 4 ;
  xa, xb := -2, 2 ;
  ya, yb := 0, 14 ;
```

$$p := x \to 1 - x^2$$

$$q := x \to -1$$

$$r := x \to x - 4$$

$$xa, xb := -2, 2$$

$$ya, yb := 0, 14$$

Differenze finite:

```
> n := 4 ;
  h := (xb-xa) / n ;
  x[0] := xa ;
  x[1] := xa + h ;
  x[2] := xa + 2*h ;
  x[3] := xa + 3*h ;
  x[4] := xa + 4*h ;
```

$$n := 4$$

$$h := 1$$

$$x_0 := -2$$

$$x_1 := -1$$

$$x_2 := 0$$

$$x_3 := 1$$

$$x_4 := 2$$

```
> eq := k -> (y[k+1]-2*y[k]+y[k-1])/h^2 +
            p(x[k]) * (y[k+1]-y[k-1])/(2*h) +
            q(x[k]) * y[k] - r(x[k]) ;
```

$$eq := k \to \frac{y_{k+1} - 2\,y_k + y_{k-1}}{h^2} + \frac{p(x_k)\,(y_{k+1} - y_{k-1})}{2\,h} + q(x_k)\,y_k - r(x_k)$$

Equazioni risultanti

```
> eq1 := eq(1) ;
  eq2 := eq(2) ;
  eq3 := eq(3) ;
```

$$eq1 := y_2 - 3\,y_1 + y_0 + 5$$

$$eq2 := \frac{3}{2}\,y_3 - 3\,y_2 + \frac{1}{2}\,y_1 + 4$$

$$eq3 := y_4 - 3\,y_3 + y_2 + 3$$

Estraggo il sistema lineare dalle equazioni

```
> A := linalg[genmatrix]([eq1,eq2,eq3],[y[1],y[2],y[3]],'b'):
  A := convert(A,Matrix);
```

$$A := \begin{bmatrix} -3 & 1 & 0 \\ \frac{1}{2} & -3 & \frac{3}{2} \\ 0 & 1 & -3 \end{bmatrix}$$

```
> b := Transpose(convert(b,Vector));
```

$$b := \begin{bmatrix} -y_0 - 5 \\ -4 \\ -y_4 - 3 \end{bmatrix}$$

Sostituisco le condizioni al contorno

```
> b := subs(y[0]=ya, y[4]=yb, b) ;
```

$$b := \begin{bmatrix} -5 \\ -4 \\ -17 \end{bmatrix}$$

Risolvo il sistema lineare

```
> res := LinearSolve(A,b) ;
  evalf(res) ;
```

$$res := \begin{bmatrix} \frac{25}{7} \\ \frac{40}{7} \\ \frac{53}{7} \end{bmatrix}$$

$$\begin{bmatrix} 3.571428571 \\ 5.714285714 \\ 7.571428571 \end{bmatrix}$$

Esercizio 4

```
> restart:
```

```
> # Punti di interpolazione
```

```
> sol := x -> x^3-x ;
```

$$sol := x \to x^3 - x$$

```
> X0,X1,X2,X3,X4 := -1,0,1,2,3 ;
  Y0,Y1,Y2,Y3,Y4 := sol(X0),sol(X1),sol(X2),sol(X3),sol(X4) ;
```

$$X0, X1, X2, X3, X4 := -1, 0, 1, 2, 3$$

$$Y0, Y1, Y2, Y3, Y4 := 0, 0, 0, 6, 24$$

```
> # condizioni al contorno
  S2_init  := (D@@2)(sol)(X0) ;
  S2_final := 24 ; # (D@@2)(sol)(X4) ;
```

$$S2\_init := -6$$

$$S2\_final := 24$$

```
> # Equazioni dei momenti
  # notate che h[k] = 1 costante perche i punti sono
  # equispaziati
  EQ1      := M0*(1/2) + 2*M1 + M2*(1/2) = 3*((Y2-Y1)-(Y1-Y0)) ;
  EQ2      := M1*(1/2) + 2*M2 + M3*(1/2) = 3*((Y3-Y2)-(Y2-Y1)) ;
  EQ3      := M2*(1/2) + 2*M3 + M4*(1/2) = 3*((Y4-Y3)-(Y3-Y2)) ;
  EQINIT   := M0 = S2_init ;
  EQFINAL  := M4 = S2_final ;
```

$$EQ1 := \frac{1}{2} M0 + 2\, M1 + \frac{1}{2} M2 = 0$$

$$EQ2 := \frac{1}{2} M1 + 2\, M2 + \frac{1}{2} M3 = 18$$

$$EQ3 := \frac{1}{2} M2 + 2\, M3 + \frac{1}{2} M4 = 36$$

$$EQINIT := M0 = -6$$

$$EQFINAL := M4 = 24$$

```
> RES := solve({EQ1,EQ2,EQ3,EQINIT,EQFINAL},{M0,M1,M2,M3,M4}) ;
```

$$RES := \left\{ M0 = -6,\ M4 = 24,\ M3 = \frac{291}{28},\ M1 = \frac{-3}{28},\ M2 = \frac{45}{7} \right\}$$

```
> # Calcolo i tratti di cubica
  S1 := Y0 + (Y1-Y0 - (M1+2*M0) / 6) * (x-X0)
          + (M0/2)*(x-X0)^2
          + ((M1-M0)/6)*(x-X0)^3 ;
  expand(subs(RES,S1)) ;
```

$$S1 := \left(-\frac{1}{6} M1 - \frac{1}{3} M0 \right) (x+1) + \frac{1}{2} M0 (x+1)^2 + \frac{1}{6} (M1 - M0) (x+1)^3$$

$$-\frac{29}{28} x - \frac{3}{56} x^2 + \frac{55}{56} x^3$$

```
> S2 := Y1 + (Y2-Y1 - (M2+2*M1) / 6) * (x-X1)
          + (M1/2)*(x-X1)^2
          + ((M2-M1)/6)*(x-X1)^3 ;
  expand(subs(RES,S2)) ;
```

$$S2 := \left(-\frac{1}{6} M2 - \frac{1}{3} M1 \right) x + \frac{1}{2} M1 x^2 + \frac{1}{6} (M2 - M1) x^3$$

$$-\frac{29}{28} x - \frac{3}{56} x^2 + \frac{61}{56} x^3$$

```
> S3 := Y2 + (Y3-Y2 - (M3+2*M2) / 6) * (x-X2)
          + (M2/2)*(x-X2)^2
          + ((M3-M2)/6)*(x-X2)^3 ;
  expand(subs(RES,S3)) ;
```

$$S3 := \left(6 - \frac{1}{6} M3 - \frac{1}{3} M2 \right) (x-1) + \frac{1}{2} M2 (x-1)^2 + \frac{1}{6} (M3 - M2) (x-1)^3$$

$$-\frac{65}{28} x + \frac{3}{7} + \frac{69}{56} x^2 + \frac{37}{56} x^3$$

```
> S4 := Y3 + (Y4-Y3 - (M4+2*M3) / 6) * (x-X3)
          + (M3/2)*(x-X3)^2
          + ((M4-M3)/6)*(x-X3)^3 ;
  expand(subs(RES,S4)) ;
```

$$S4 := 6 + \left(18 - \frac{1}{6} M4 - \frac{1}{3} M3 \right) (x-2) + \frac{1}{2} M3 (x-2)^2 + \frac{1}{6} (M4 - M3) (x-2)^3$$

$$-\frac{87}{7} + \frac{475}{28} x - \frac{471}{56} x^2 + \frac{127}{56} x^3$$

```
>
>
```